import std.conv, std.stdio;
import std.algorithm, std.array, std.range, std.string;
import std.functional;
void main()
{
    immutable inserts = readln.chomp.to!int;
    immutable original = readln.chomp.length;
    (inserts + 1).iota.map!(i => c(inserts+original, i) * Mod!10_0000_0007(25) ^^ i).sum.writeln;
}

private:

Mod!10_0000_0007 c(in ulong n, in ulong r)
{
    if (r == 0)
        return Mod!10_0000_0007(1);
    return memoize!c(n, r-1) * (n-r+1) / r;
}

struct Mod(ulong mod = 10_0000_0007)
{
    long v;
    auto opOpAssign(string op)(in long rhs)
        if (op == "+" || op == "*")
    {
mixin ("v "~op~"= rhs;");
        v %= mod;
        return this;
    }
    auto opOpAssign(string op)(in Mod!mod rhs)
        if (op == "+" || op == "*")
    {
        return opOpAssign!op(rhs.v);
    }
    auto opBinary(string op, T)(in T rhs) const
        if ((op == "+" || op == "*") &&
            (is (T : long) || is (T == Mod!mod)))
    {
        Mod!mod lhs = this;
        return lhs.opOpAssign!op(rhs);
    }
    auto opBinary(string op)(in long rhs) const
        if (op == "/")
    {
        return this * rhs.modinv!mod;
    }
    auto opBinary(string op)(in long rhs) const
        if (op == "^^")
    {
        if (rhs == 0)
            return Mod!mod(1);
        if (rhs == 1)
            return this;
        auto sq = this * this;
        return sq ^^ (rhs >> 1) * this ^^ (rhs & 1);
    }
    string toString() const
    {
        return "%d".format(v);
    }
}

/// from https://qiita.com/drken/items/3b4fdf0a78e7a138cd9a#3-5-%E6%8B%A1%E5%BC%B5-euclid-%E3%81%AE%E4%BA%92%E9%99%A4%E6%B3%95%E3%81%AB%E3%82%88%E3%82%8B%E9%80%86%E5%85%83%E8%A8%88%E7%AE%97
ulong modinv(long mod)(long a)
{
    long b = mod, u = 1, v = 0;
    while (b)
    {
        immutable q = a / b;
        a -= q * b;
        a.swap(b);
        u -= q * v;
        u.swap(v);
    }
    u %= mod;
    if (u < 0)
        u += mod;
    return u;
}

unittest
{
    alias minv = modinv!10_0000_0007;
    foreach (i; 1..1000)
    {
        assert (i * minv(i) % 10_0000_0007 == 1, "1 / %d != %d".format(i, minv(1)));
    }
}
